
1      page 40,132
2     ;_______________________________________________________________________________
3     ; Tiny Boot Select (TBOOTSEL). Copyright (C) 2016-2021 Ton Daas
4     ; TBOOTSEL is free software: you can redistribute it and/or modify
5     ; it under the terms of the GNU General Public License as published by
6     ; the Free Software Foundation, either version 3 of the License,
7     ; or any later version.
8     ; TBOOTSEL is distributed in the hope that it will be useful,
9     ; but WITHOUT ANY WARRANTY; without even the implied warranty of
10     ; MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.
11     ; See the GNU General Public License for more details.
12     ;
13     ; You should have received a copy of the GNU General Public License
14     ; along with this program.  If not, see <https://www.gnu.org/licenses/>.
15     ;_______________________________________________________________________________
16     ; Tiny Boot Select is a manager that resides entirely in the master boot record.
17     ; It allows the user to choose from up to four partitions he wants to boot from,
18     ; or revert to PC-BIOS for starting ROM-Basic or a subsequent boot device.
19     ; Dos/Win partitions can be hidden, to allow for multiple primary partitions.
20     ; The selected partition will be unhidden, hiding other conflicting partitions.
21     ; It supports FAT12, FAT16, NTFS, FAT32, partitions >2GB and disks >8GB, <2TB.
22     ; It is a modified version of Tiny Boot Manager, to suit MS Windows 7-10.
23     ; This means that the master boot record will be saved with the last selected
24     ; partition marked active. Consequently the user prompt will always be shown and
25     ; unused table entries and an extended partition are treated as invalid choices.
26     ;===============================================================================
27  0000    tbootsel segment
28      assume cs:tbootsel,ds:tbootsel
29  7C00     org 7C00h
30  = 7C00    loadadr equ $
31  0600     org 600h
32     
33     ; Master boot record routine at CYL=0, HEAD=0, SECT=1
34     ; Upon execution reg values are: CS=0000h, IP=7C00h, DL=drive
35     ; Initialize stack and relocate code from 7C00h to this address at 0600h

36  0600    mbr proc near
37  0600  FC    cld
38  0601  33 C9    xor cx,cx
39  0603  8E D9    mov ds,cx ;set source segment
40  0605  BE 7C00 R   mov si,offset loadadr ;set source offset loaded code
41  0608  8E C1    mov es,cx ;set destination segment
42  060A  BF 0600 R   mov di,offset mbr ;and target offset
43  060D  FA    cli  ;prevent interrups prior to changing stack location
44  060E  8E D1    mov ss,cx ;set stack segment
45  0610  8B E6    mov sp,si ;set stack to before source area
46  0612  B5 01    mov ch,1h ;256 words
47  0614  F3/ A5    rep movsw ;move this code from address 7C00h to 600h
48  0616  FB    sti  ;allow interrups after string move
49  0617  E9 9045 R   jmp near ptr cont-loadadr+mbr ;near jump will do nicely
50     
51     ; Ascii special characters
52  = 0007    bel equ 07h
53  = 000A    lf equ 0Ah
54  = 000D    cr equ 0Dh
55  = 001B    escape equ 1Bh
56     
57     ; Recognized partition type list
58  = 0001    fat12 equ 01h ;11h if hidden, <32Mb
59  = 0004    fat16 equ 04h ;14h if hidden, >32Mb <500Mb
60  = 0005    extend equ 05h ;extended
61  = 0006    fat16b equ 06h ;16h if hidden, >32Mb <2Gb
62  = 0007    ntfs equ 07h ;17h if hidden
63  = 000B    fat32 equ 0Bh ;1Bh if hidden, <2Gb
64  = 000C    f32lba equ 0Ch ;C/H/S=unused, relative sector=LBA
65  = 000E    f16lba equ 0Eh ;C/H/S=unused, relative sector=LBA
66  = 000F    extlba equ 0Fh ;C/H/S=unused, relative sector=LBA
67     
68  = 0010    hidden equ 10h ;single bit is set in DOS partition types 
69  = 0080    bootflg equ 80h
70     

71     ; Types that have hidden equivalent
72  061A  01 04 06 0B  dostype db fat12,fat16,fat16b,fat32
73  = 0004    chslen equ $-dostype ;dos types that use chs method
74  061E  0E 0C    db f16lba,f32lba ;dos types that use lba method
75  0620  07    db ntfs  ;type supports both
76  = 0007    typelen equ $-dostype
77     
78     ; Routine to read or write 1 sector
79     ; AH= function (2-read or 3-write)
80  0621  B0 08   int13: mov al,100h shr 5 ;initialize retry count to 5
81  0623  8B F8   int13r: mov di,ax
82  0625  B0 01    mov al,1 ;specify one sector transfer
83  0627  CD 13    int 13h
84  0629  73 09    jnc ret13
85  062B  B4 00    mov ah,0 ;reset disk system
86  062D  CD 13    int 13h
87  062F  97    xchg ax,di
88  0630  D0 E0    shl al,1
89  0632  73 EF    jnc int13r ;try al shift times
90  0634  C3   ret13: ret
91     
92     ; Routine to write text to screen; exits with ds:si pointing at 0
93     ; Entrypoint is tty (or wrtty if AL already has first character to write)
94     ; Entrypont is ttyeol to output only AL followed bij end of line
95     ; On entry ds:si points to message to write (null terminates routine)
96  0635  BE 07AC R  ttyeol: mov si,offset crlf
97  0638  B3 07   wrtty: mov bl,7 ;white
98  063A  B4 0E    mov ah,0Eh ;write teletype to active page
99  063C  CD 10    int 10h ;AL=character, BL=foreground color
100  063E  AC   tty: lodsb
101  063F  3C 00    cmp al,0
102  0641  75 F5    jnz wrtty ;end on nul
103  0643  4E    dec si ;set pointer back to trailing 0
104  0644  C3    ret
105     

106     ; Look for an active partition
107  0645  BF 0040   cont: mov di,4*16 ;initialize as invalid at end of table
108  0648  8D 75 F0   lea si,[di-16] ;set search index at last entry
109  064B  B8 0080    mov ax,bootflg ;load valid bootflag, clear AH
110  064E  86 A4 07BE R  chk_bf: xchg ah,table[si] ;get bootflag in AH, clear in table
111  0652  84 E4    test ah,ah ;check for empty bootflag field
112  0654  74 07    jz next ;active flag clear?
113  0656  32 C4    xor al,ah ;check valid bootflag and make future flags invalid
114  0658  75 0E    jnz inval ;invalid bootflag field?
115  065A  8B FE    mov di,si ;save current active table entry offset
116  065C  98    cbw  ;clear ah
117  065D  83 EE 10  next: sub si,16
118  0660  73 EC    jnc chk_bf ;not all 4 partitions done?
119  0662  BE 0757 R   mov si,offset prompt
120  0665  AC    lodsb
121  0666  EB 10    jmp short rdkey ;go and prompt user for selection
122     
123  0668  BE 0776 R  inval: mov si,offset inv_msg
124  066B  E9 072A R   jmp abend
125     
126     ; Exit to BIOS to start ROM Basic or subsequent boot device
127  066E  E8 0635 R  basic: call ttyeol ;show entered keystroke in AL to user
128  0671  CD 18    int 18h ;start ROM Basic
129     
130     ; Prompt user to select a partition
131  0673  BE 07AE R  reask: mov si,offset null
132  0676  B0 07    mov al,bel ;load bel char
133  0678  E8 0638 R  rdkey: call wrtty
134  067B  B4 00    mov ah,0 ;read keyboard
135  067D  CD 16    int 16h ;AL=ascii, AH=scancode
136  067F  3C 1B    cmp al,escape ;Esc char
137  0681  74 EB    je basic ;esc key pressed?
138  0683  8B E8    mov bp,ax ;save keystroke
139  0685  2C 31    sub al,'1' ;convert numeric 1 base ascii to binairy 0 base
140  0687  A8 FC    test al,not 3h ;if not within 0 to 3,

141  0689  75 E8    jnz reask ;invalid choice, reask user
142  068B  B4 10    mov ah,16
143  068D  F6 E4    mul ah ;convert to table entry offset
144  068F  96    xchg si,ax ;transfer offset to SI
145     
146     ; Is selection unused or an extended partition?
147  0690  BB 07BE R   mov bx,offset table
148  0693  8A 40 04   mov al,[si+bx+4] ;get partition type
149  0696  84 C0    test al,al  ;if it is an unused entry?
150  0698  74 D9    jz reask  ;then reask
151  069A  3C 05    cmp al,extend ;if extended dos partition choosen?
152  069C  74 D5    je reask ;then reask
153  069E  3C 0F    cmp al,extlba ;if extended lba dos partition choosen?
154  06A0  74 D1    je reask ;then reask
155     
156     ; Mark selected partition active
157  06A2  C6 00 80   mov byte ptr [si+bx],bootflg
158  06A5  3B F7    cmp si,di ;if selection is same as prevous,
159  06A7  8D 30    lea si,[si+bx]
160  06A9  74 34    je no_chg ;then no need to update MBR
161     
162     ; Unhide partition if selected partition is hidden dostype
163  06AB  34 10    xor al,hidden ;unhide any hidden partition
164  06AD  BF 061A R   mov di,offset dostype
165  06B0  B9 0007    mov cx,typelen
166  06B3  F2/ AE    repne scasb
167  06B5  75 1C    jne wrtchg ;selected partition other than hidden dostype?
168     
169     ; Hide any other unhidden dos partition
170  06B7  3B DE   hide: cmp bx,si
171  06B9  74 0C    je unhide ;selected partition?
172  06BB  8A 47 04   mov al,[bx+4]
173  06BE  BF 061A R   mov di,offset dostype
174  06C1  B1 07    mov cl,typelen
175  06C3  F2/ AE    repne scasb

176  06C5  75 04    jne skphid ;already hidden or non dos?
177  06C7  80 77 04 10  unhide: xor byte ptr[bx+4],hidden ;hide partition
178  06CB  8D 5F 10  skphid: lea bx,[bx+16]
179  06CE  80 FB FE   cmp bl,low offset table+64
180  06D1  72 E4    jb hide
181     
182     ; Write changed partition table back to master boot record
183  06D3  BB 0600 R  wrtchg: mov bx,offset mbr ;buffer address
184  06D6  B6 00    mov dh,0 ;set head=0, drive number is still in DL
185  06D8  B1 01    mov cl,1 ;set sector=1 and cylinder=0 (assume CH=0)
186  06DA  B4 03    mov ah,3 ;write back master boot record
187  06DC  E8 0621 R   call int13 ;continue even with error
188     
189     ; Output selection to user
190  06DF  56   no_chg: push si ;save si
191  06E0  95    xchg ax,bp ;reload keystroke and save table entry in BP
192  06E1  E8 0635 R   call ttyeol ;show selection choice
193  06E4  5E    pop si ;restore si
194     
195     ; Analize partition type
196     ; mov al,[si+4] ;get partition type
197     ; mov di,offset dostype
198     ; mov cl,chslen
199     ; repne scasb ;is active partition chs type?
200     ; je rdchs ;read partition traditional chs style
201     
202     ; For all other types check for extended int 13h support
203  06E5  BB 55AA    mov bx,55AAh ;fill with request signature
204  06E8  B4 41    mov ah,41h ;get extended int 13 support info; DL still has drive
205  06EA  CD 13    int 13h
206  06EC  72 41    jc rdchs ;extension not found
207  06EE  81 FB AA55   cmp bx,0AA55h ;signature, AH=major version, DH=extension ver.
208  06F2  75 3B    jne rdchs ;requested support not installed
209  06F4  F6 C1 01   test cl,01h ;bit0=1 if int13,AH=42h supported
210  06F7  74 36    jz rdchs ;API subset not supported

211     
212     ; Read bootrecord with extended interrupt 13h
213  06F9  8B DC    mov bx,sp ;get bootsector load address
214  06FB  B9 0005    mov cx,5 ;set retrycount
215  06FE  56   retrlb: push si ;save si
216  06FF  33 C0    xor ax,ax
217     ; Build address request packet
218  0701  50    push ax ;sector 4th word
219  0702  50    push ax ;sector 3rd word
220  0703  FF 74 0A   push [si+10] ;sector 2nd word
221  0706  FF 74 08   push [si+8] ;sector low word
222  0709  06    push es ;buffer segment
223  070A  53    push bx ;buffer offset
224  070B  40    inc ax
225  070C  50    push ax ;number of sectors (max.(7F)
226  070D  B0 10    mov al,10h ;packet size
227  070F  50    push ax ;high byte reserved (=0)
228  0710  8B F4    mov si,sp ;DS:SI points to request address packet
229  0712  B4 42    mov ah,42h ;extended disk read; DL has drive number
230  0714  CD 13    int 13h
231  0716  72 04    jc skp_ck ;if CF then AH=errorcode else AH=0
232     ; Check sector count read, as CF is not set if sector not found error
233  0718  83 7C 02 01   cmp word ptr[si+2],1 ;also need to check actual count
234  071C  8D 64 0E  skp_ck: lea sp,[si+14] ;purge address request packet from stack
235  071F  58    pop ax ;and last word of package into AX
236  0720  5E    pop si ;restore si
237  0721  73 1B    jnc readok
238  0723  CD 13    int 13h ;reset disk system
239  0725  E2 D7    loop retrlb
240  0727  BE 078E R  rdfail: mov si,offset err_msg
241  072A  E8 063E R  abend: call tty
242  072D  EB FB    jmp short abend ;loop on last 0
243     
244     ; Read bootrecord of active partition
245  072F  8B DC   rdchs: mov bx,sp ;set ES:BX to buffer address 7C00h

246  0731  8A 74 01   mov dh,[si+1] ;set head number, DL still has drive number
247  0734  8B 4C 02   mov cx,[si+2] ;set sector & cyl
248  0737  B4 02    mov ah,2 ;read partition bootsector
249  0739  E8 0621 R   call int13
250  073C  72 E9    jc rdfail
251  073E  81 BF 01FE AA55  readok: cmp word ptr [bx+(bootid-mbr)],0AA55h
252  0744  75 02    jne noboot
253  0746  FF E3    jmp bx ;execute partitions boot record
254     ; Upon exit DS:SI points to booted partition table entry
255     
256  0748  BE 0794 R  noboot: mov si,offset mis_msg
257  074B  8B FE    mov di,si
258  074D  B8 694D    mov ax,'iM'
259  0750  AB    stosw  ;replace first characters with 'Miss'
260  0751  B8 7373    mov ax,'ss'
261  0754  AB    stosw
262  0755  EB D3    jmp short abend
263     
264  0757  53 74 61 72 74 20 prompt db 'Start partition (1-4 or Esc)?:',0
265        70 61 72 74 69 74
266        69 6F 6E 20 28 31
267        2D 34 20 6F 72 20
268        45 73 63 29 3F 3A
269        00   
270  0776  49 6E 76 61 6C 69 inv_msg db 'Invalid partition table',0
271        64 20 70 61 72 74
272        69 74 69 6F 6E 20
273        74 61 62 6C 65 00
274  078E  45 72 72 6F 72 20 err_msg db 'Error ' ;loading operating system'
275  0794  6C 6F 61 64 69 6E mis_msg db 'loading operating system' ;first 4 characters could be patched
276        67 20 6F 70 65 72
277        61 74 69 6E 67 20
278        73 79 73 74 65 6D
279  07AC  0D 0A   crlf db cr,lf ;should be terminated by zero
280  07AE     07 [   null db 1B5h-($-mbr) dup (0) ;fill unused space

281                 00  
282                     ]  
283     
284  07B5  76 8E 94   db low offset inv_msg, low offset err_msg, low offset mis_msg
285  07B8     06 [    db 6 dup (0) ;Windows NT signature
286                 00  
287                     ]  
288     
289  07BE     40 [   table db 64 dup (0)
290                 00  
291                     ]  
292     
293  07FE  AA55   bootid dw 0AA55h
294  0800    mbr endp
295  0800    tbootsel ends
296      end


